
· Screens are made up of more than just a screen layout with input and output fields. They also have their own processing logic. 

· The fact that the ABAP Dictionary is integrated in the system means that automatic consistency checks for screen input fields are provided. These checks include type checks, foreign key checks, and fixed value checks. All of these checks are automatically supplied with information from the ABAP Dictionary.

· Checks like the ones above can be complemented by other program-specific checks. With the techniques available for screens , you can control the order in which checks are performed and, if errors occur, make the fields input ready again, where appropriate.

· You can also set up screen layout in a very flexible way. Input fields, output fields, radio buttons, checkboxes, and even pushbuttons can be placed on screens. These features allow users to determine the direction in which the program will proceed.

· The same editing options are available for the screen as for list display and the selection screen: the fixed point numbers and the date are set according to the user default values, the time in format hh:mm:ss, amounts according to the setting in a currency field, and lengths and weights according to the content of a unit-of-measure field.

· To start a screen sequence:

· Specify the first screen as the start screen of a transaction code of the type dialog transaction
· Call a screen from any one ABAP processing block in the program

· This unit concentrates on the latter option.

· In the following units, you will develop a program, in several steps, that enables the user to change the master data of the flight timetable. 

· Double-clicking on an entry in the basic list containing the flight timetable calls a screen, which displays data from the line you selected, and additional information about the flight connection. You can change the flight duration and the departure time.

· Choosing Back takes the user back to the basic list without changing any data.

· Choosing Save changes the data in the database.

· There is a function module for database changes that is implemented later on (see the unit about Cross-Program Modularization).

· Each screen has the following information:

· Attributes:
Include a four-digit number as the screen name, designation, and a short text information on the screen type (for example, normal for full size screen). 

· Layout:
Includes elements you can place on the screen. Elements that are displayed on the screen are called screen elements.

· Element list:
Contains the attributes of the screen elements, such as the position, size, and data type. 

· Flow control:
Contains the processing logic that is to be processed before the screen is sent to the presentation server (PBO) and the processing logic that is to be processed after a user action has taken place.

· From the Graphical Layout Editor, you can start other functions: 

· Element attributes:
All the attributes of a screen element are displayed in this dialog box. You can change some of the attributes in the box. Example: you can specify whether or not an input/output field is to be input ready.

· Get from ABAP Dictionary / Get from program:
This dialog box allows you to generate fields that have either a global type or the same type as a data object in the program.

· Element list:
Shows all the elements displayed on the screen with their attributes. You can also change attributes here..

· Your first step is to create a screen, specify its layout, and define its field attributes.
The fields Airline, Flight number, Departure airport, and Destination airport are to appear as output fields. The fields Flight duration and Departure time are available as input-ready fields. 

· The user should be able to call the screen by double-clicking a line within the basic list and return to the basic list by choosing the appropriate function on the screen.

· There are several ways to create screens:

· Through forward navigation: You can create screens from within the ABAP Editor by double-clicking on the screen number. This transfers you to Screen Painter automatically.

· Using Object Navigator: You can create a new program object screen for your program directly from the object list in the navigation area.

· When you create a screen, the system will ask you to enter screen attributes. Enter a short description of the screen, select screen type Normal, and enter the number of the subsequent screen.

· If you enter 0 for the subsequent screen, the system first processes your screen completely and then returns to processing at the point where the screen call is set. (Note that "0" in the entry field for the subsequent dynpro is not displayed because this is actually the inital value.)

· In the example in the graphic, the screen the user creates is called from within the basic list. Therefore, CALL SCREEN 100 must belong to the AT LINE-SELECTION event block.

· There are two ways of assigning field attributes to the required screen fields:

· Copying data from the Dictionary: You can copy types and field attributes from existing ABAP Dictionary structures. This makes all information about the object available to you, including semantic information about its data elements and foreign key dependencies. The name of the Dictionary field is automatically adopted as a field name.

· Copying data from the program: You can copy field attributes from data objects already defined within your program to the screen. In order to do this, however, an activated copy of the program must already exist. The name of the data object is automatically taken as the field name.

· The Graphical Screen Painter allows you to define different screen elements (for example, input and output fields, keyword texts, borders, and so on) with relative ease. Choose the desired screen element and then place it on the screen using the mouse.

· You can delete screen elements simply by pointing to them and choosing Delete.

· You can move screen elements by selecting them and dragging them to a new position.

· To maintain the attributes of a screen field, select the field and choose Attributes. 

· You can assign the attribute mandatory (Required field) to a screen field. At run time, the field will be marked accordingly if it is set to initial value.

· If not all required fields have been filled at run time and a user action is performed, an error dialog is triggered by the ABAP run time system and all input fields are once again displayed ready for input.

· In step two you learn how to display data on a screen by programming data transport from the basic list to the screen. 

· For the user, the program works in the following manner:

· By double-clicking on a line in the basic list, the user branches to the screen where the most important items of information for the flight connection he or she has chosen are displayed. The flight duration and departure time are displayed in a field that is ready for input and hence can be changed.

· The user returns to the basic list in one of several ways.

· With this in mind, this part of the unit will deal with:

· Setting up prerequisites for automatic data transport between the program and screen fields

· Defining the screen interface and programming data transport to the data objects of the interface

· The TABLES statement defines an internal data object that serves as an interface for the screen. The TABLES statement always refers to an ABAP Dictionary object (for example, transparent table or structure). 

· If the screen fields and the TABLES statement refer to the same ABAP Dictionary object, this data object's data is automatically transported to the screen fields every time the screen is transmitted. Any new entries or changes that the user makes on the screen are then transferred back into this data object.

· As a rule, structures that contain fields of the same type for the different tables are created in the ABAP Dictionary. The flight data programs being created in this course use one structure for master data maintenance (sdyn_conn) and another for bookings data (sdyn_book). Using your own structures as interfaces usually helps make a program easier to understand and helps to avoid errors as well.  

· Data transport between program data objects and screen fields with the same name takes place automatically:
Immediately before the screen is sent to the presentation server, that is, after the PBO event has been fully processed, the contents from fields with the same names are copied from the ABAP work area into the screen fields. 

· ABAP statements facilitate data transport between ABAP program data objects and the work area designated as the screen interface.

· Data transport between screen fields and program data objects with the same name takes place automatically: 
Immediately after a user performs an action on the screen, that is, before the PAI event is fully processed, the contents of the fields with the same name are copied from the screen fields into the ABAP work area. 

· ABAP statements facilitate data transport between jects the work area designated as the screen interface and program data objects..

[image: image16.wmf]ã

SAP AG 2001

Data

Transport in

the Program Example

ABAP

runtime system

CALL SCREEN 100.

Time

Automatic

data transport

HIDE

area

in

data objects

TABLES

structure

in

screen fields

Screen fields

in

TABLES

structure

AT LINE

-

SELECTION.

Fill the 

TABLES

structure 

with current data

for selected line


· The sample program should display screen data that matches the line selected in the basic list. 

· If data objects and their values were stored in the HIDE area when the basic list was created, the data belonging to the selected line will be placed in the corresponding data objects.

· You must copy the data to be displayed on the screen to a TABLES structure. You can do this in any ABAP processing block that is processed before the screen is sent to the presentation server. There are various ways of doing this:

· You start by reading all the data to be displayed on the screen before retrieving the basic list from the database. You then place all the necessary data in the HIDE area. Then, at the AT LINE-SELECTION event, you have to copy the data into the TABLES structure only. 
Disadvantage: You have to read data from the database that the user may not even look at. If detailed data has changed between creating the basic list and displaying the screen, the system will display the wrong data.

· Before the screen is called using the SELECT SINGLE statement, you read the additional data for the selected key (HIDE area) from the database.
Advantage: You reduce the volume of data you need to read from the database when you create the basic list. The detailed list is up-to-date.
Disadvantage: The system sends a query to the database every time the user double-clicks the list.

· On the last development level, the program should allow the user to change data in the database. The user should be able to change the fields FLTIME and DEPTIME. To enable the user to change data for several airlines, a basic list of the airlines for which the user is allowed to change data should be displayed. The user reaches the change screen by double-clicking in the basic list. Once the changes are made successfully, the user returns to the basic list. However, a new basic list is not created. Therefore, the data that can be changed should not appear on the basic list. 

· To ensure that the database data displayed on the screen is up-to-date, the record is read again from the database at the beginning of AT LINE-SELECTION. 

· Advantages of this method:

· For the basic list, only those columns of the database table that are displayed on the list need to be read. This can improve performance with large lists. 

· The data that is displayed on the screen is always up-to-date, even if the data record selected has just been changed using this program. 

· Changes made to the database using the screen do not lead to incorrect values in the basic list because the modifiable fields are not contained in the basic list. 

· To display data on the screen, the TABLES structure must be filled with current data before the screen is sent to the presentation server. The example in the graphic shows one way of doing this.

· The HIDE statement is used to place key fields of database tables with reference to the list line in the HIDE area. Then the current data for the line selected is available in fields wa_spfli-carrid and wa_spfli-connid at event AT LINE-SELECTION.

· The data record is read from the database using SELECT SINGLE. This ensures that the data is up-to-date. 

· The corresponding fields of the wa_spfli work area are copied to the TABLES structure sdyn_conn using MOVE-CORRESPONDING. The system transports the structure data to the screen fields automatically.

· Alternatively, you can place the data into the TABLES structure directly when the database is accessed, using the INTO CORRESPONDING FIELDS addition.

· In step three you learn how to assign pushbutton functions. These functions allow different kinds of program logic to be processed according to user choice. 

· For the user, the program works in the following manner:

· By double-clicking on a line, the user branches to a screen where the most important items of information for the connection the user chose are displayed. The flight duration and departure time can be changed.

· By choosing the Back pushbutton, the user returns to the basic list without writing any changes to the database. The message "Screen was left without any changes being made" is displayed in the status bar of the basic list.

· Choose Save to write all of your changes to the database. The changes take effect in the program only at a later time. The pushbutton is already prepared in the following section. The user returns to the basic list after choosing the pushbutton and a message is displayed in the status bar. . 

· After each other user activity, the screen is displayed once again.

· With this in mind, this part of the unit deals with:

· Flow logic in PBO and PAI event blocks

· Using PBO and PAI modules as ABAP processing blocks for screen programming

· Evaluating information on which pushbutton was selected at the time of PAI

· If the user chooses a pushbutton, the run time system copies the associated function code to a special screen field. This screen field is usually called the ok_code.
· The content of this screen field is then automatically transported if there is an ABAP data object of the same name.

· The content of this ok_code field can then be evaluated in an ABAP processing block.

· This mechanism allows you to create a program flow that depends on the user's actions. The following slides deal with how you declare the ok_code field, how you create pushbuttons and assign function codes to them, and how you can change the screen sequence.  

· To be able to transport information as to which pushbutton was chosen, you must assign a function code to each pushbutton. You can do this either in the Graphical Layout Editor using the maintenance function for field attributes or from the element list.

· The command field serves as a special field into which corresponding function codes are placed after every user action.

· This field must be given a name in the last line of the list of elements on each screen. Generally, you use the name OK_CODE.

· In the declaration part of the program, define a corresponding data object with the same name. At run time, whenever a user action is triggered, the function code of the pushbutton chosen is placed into the data object. You can use the system field sy-ucomm as a reference field for the type.

· The ABAP statement CALL SCREEN <nnnn> interrupts the processing of the processing block and calls a screen. 

· Each screen has two event blocks:

· PROCESS BEFORE OUTPUT (PBO)  is processed immediately before a screen is displayed. At this time, modules are called that take care of tasks such as presetting values in screen fields.

· PROCESS AFTER INPUT (PAI) is processed immediately after a user action. All program logic that is influenced by user action must be processed at PAI.   

· Note: The code for the events PBO and PAI is written using the Screen Painter, not with the ABAP Editor. These two event blocks make up the flow logic of a screen. A special small set of commands (not ABAP statements) is available for programming flow logic. The most important statement here is called: MODULE module_name. This calls the module_name ABAP processing block. Within this block you can code in ABAP as usual.  

· Modules are ABAP processing blocks with no interface that can be called only from within the flow logic. Modules are enclosed within the ABAP statements MODULE and ENDMODULE.

· You can evaluate user actions in a PAI module. Usually, this PAI module is called user_command_nnnn, where nnnn stands for the screen number. In this module, evaluate the function code in the ok_code field.

· Note: For historical reasons, modules have no interface and no local variables. You can access all the ABAP program's global data in modules.

· As a rule, you implement MODULE calls within a screen's flow control (PBO and PAI events). The modules themselves are created using ABAP. 

· There are two ways to create a module:

· Forward navigation: To create a module, double-click on the module name in the flow logic from within the Screen Painter.

· Navigation area: If you want to create a module from within the object list of the program, choose a new program object  PBO module or PAI module.  

· A module can be called from more than one screen (reusability). 

· Make sure that modules called at PBO events are defined using the statement MODULE ... OUTPUT, whereas modules defined using the statement MODULE ...  INPUT can be called only at PAI events.

[image: image26.wmf]ã

SAP AG 2001

Time

Next Screen

: CALL SCREEN Statement

100

PBO

CALL SCREEN 100.

PAI

Screen attributes

Next screen

0

Next screen

0


· If you enter 0 for the next screen, the system first processes your screen completely and then continues processing the program from the point where the screen was called.

· If you set the next screen for screen 100 to 100, the system processes the screen again after it has finished processing the PAI module.

· Using the ABAP statement SET SCREEN nnnn from within a PAI module, you can have the next screen set dynamically. The static entry is then suppressed.

· The number of the current screen is entered by the system as the standard entry for the next screen. This ensures that the Enter key or the green checkmark lead to a field check only; afterwards the screen is transmitted once again, possibly with a respective error message. To leave the screen, an appropriate pushbutton must be defined that triggers a screen change within the PAI module.

· Note: If the system processes the same screen again, it also runs through all the PBO modules again. If you decide to fill the TABLES structure in a PBO module, make sure that data changes made by the user are not overwritten on the screen if the module gets called twice

ENDMODULE.

.

.

· For the example for evaluating function codes, two pushbuttons are discussed:

· Choosing BACK dynamically sets this value to 0. This sends the user back to the callpoint. In our case, this means the user returns to the basic list. Message 057 appears in the status bar of the next screen.

· With SAVE, the program behaves in the same way as BACK. The user returns to the basic list; however, there is a new message in the status bar. In the unit about Cross-Program Modularization, the program is enhanced with a data record change.

· This section walks through a particular scenario to point out the necessity of an additional program step.

· The user starts the program and double-clicks to display detailed information on the screen. After determining that all the data is correct the user returns to the basic list by choosing the green arrow. In this way, the function code BACK, which is assigned to this standard pushbutton, is placed in the command field and then it is transported automatically to the global data object ok_code of the ABAP program. This function code is then evaluated in a PAI module. The next screen is set to 0.

· Then AT LINE-SELECTION is processed further. If there is no WRITE statement there, the system displays the basic list again.

· The user then displays details for another record by double-clicking it. Through automatic field transport, the content of the global data object ok_code is transported into the screen field with the same name and the screen is displayed.

· If the user now chooses Enter, the screen is displayed once again, provided no function code has been assigned to the Enter key. (Standard case!) Therefore, the command field is not overwritten. The function code BACK remains in the command field and is copied to the global data object ok_code at the beginning of the PAI event.

· This function code is then evaluated in a PAI module. Consequently, as described above, the system goes back to the basic list, instead of displaying the screen again. 

· If the command field is not initialized, errors can occur since not every pushbutton must have a function code assigned to it. There are two ways to initialize a command field:

· Initialize the ok_code field in a PBO module. Then it is set to the initial value at PAI, unless the user has carried out a user action to which a function code is assigned. In this case, the ok_code field contains the current function code.

· Use an auxiliary field and copy the content of the ok_code field into the auxiliary field in a PAI module, then initialize the ok_code field. In this case, the auxiliary field must be queried in the PAI module for the function code evaluation.

	Unit: Screen

Topic: Creating Screens

	At the conclusion of these exercises, you will be able to:

· Create screens

· Call existing screens from the program

	The program SAPBC400UDT_DYNPRO_A enables you to display all bookings made by one travel agency as a list. 
As a first step, change the program as follows:
A screen is to be called whenever the user double-clicks a line in the basic list. This screen should contain input fields for specific booking data that is not displayed on the list. This screen should also contain output fields for booking information that is already displayed on the list. Any user action should result in the basic list being displayed again

	Program:

      ZBC400_##_DYNPRO
Model solution:
SAPBC400UDS_DYNPRO_A
Template:
            SAPBC400UDT_DYNPRO_A




1-1
Copy the template SAPBC400UDT_DYNPRO_A to your new program ZBC400_##_DYNPRO.

1-2
Become familiar with the program. Test the program using the travel agency number 1## (## is your group number).

1-3
Selecting a line on the basic list (by double-clicking or using F2) should call a screen. Create this screen (screen number 100) using forward navigation.

1-4
For the attributes, assign screen number 0 as the number of the next screen, so that after any user action on screen 100, the user returns to the basic list.

1-5
Create input/output fields on the screen. When you are assigning field types, refer to ABAP ABAP Dictionary structure SDYN_BOOK.

· Copy booking table key fields CARRID, CONNID, FLDATE, and BOOKID with their field labels.

· The customer name NAME should be copied without a field label and should be displayed next to the customer number. 

· The fields CUSTOMID, CUSTTYPE, SMOKER, CLASS, LOCCURAM, and LOCCURKEY should be copied with field labels. 


1-6
Maintain the screen field attributes:

· Fields CARRID, CONNID, FLDATE, BOOKID, and CUSTOMID should be displayed as output fields (Output field attribute).

· The customer name NAME should be displayed next to the customer number without text (Output only attribute).

· The fields CUSTTYPE, SMOKER, CLASS, LOCCURAM, and LOCCURKEY are input/output fields (Input field/Output field attributes). 

	Unit: Screen

Topic: Data Transport

	At the conclusion of these exercises, you will be able to:

· Fill the screen fields with data from the program

	Change your program ZBC400_##_DYNPRO: 
Double-clicking on a line of the basic list displays details of the selected booking on the screen. If the user changes data on the screen, then these changes should be available in the program once the user has left the screen.

	Program:

ZBC400_##_DYNPRO
Model solution:
SAPBC400UDS_DYNPRO_B



2-1 Change your program ZBC400_##_DYNPRO or copy the template SAPBC400UDS_DYNPRO_A to the new program ZBC400_##_DYNPRO_B.

2-2 In your program, use the TABLES statement to create a work area with the same name. This work area will then serve as the screen interface.

2-3
Ensure that the key fields of the SBOOK database table and the customer name are still available (HIDE: ...) in the AT LINE-SELECTION event block after a line has been selected on the basic list (double click or F2).

2-4
Change your program accordingly so that data in the database can be changed. Make sure that the screen can be processed only if the user has change authorization for the airline selected.
Before the screen is called, retrieve the data of the selected posting from the SBOOK database table into a suitable work area (name proposal: wa_sbook) To make sure that the data is up-to-date. If the data record cannot be read, the system must display information message 176 from message class BC400. If the data is read successfully, call the screen.

2-5
Immediately before the screen is called, copy the relevant data into the TABLES work area.

	Unit: Screen

Topic: Field Transports and Subsequent Screen Processing

	At the conclusion of these exercises, you will be able to:

· Create pushbuttons on screens

· Process the system code triggered when the user clicks on a pushbutton and thus control the program flow

· Set the next screen dynamically

	Change your program ZBC400_##_DYNPRO:
On the screen, the user should be able to control program flow through two pushbuttons.

	Program:

ZBC400_##_DYNPRO
Model solution:
SAPBC400UDS_DYNPRO_C



3-1
Change your program ZBC400_##_DYNPRO or copy the template SAPBC400UDS_DYNPRO_B to the new program ZBC400_##_DYNPRO_C.

3-2
Define two pushbuttons on the screen that allow the user to either return to the basic list (PUSH_BACK) or to save the changed data (PUSH_SAVE):

	Name of pushbutton
	Text
	Function code

	PUSH_BACK
	Back
	BACK

	PUSH_SAVE
	Save
or icon
ICON_SYSTEM_SAVE
	SAVE


3-3
Name the OK_CODE field on the screen and define a data object of the same name (and corresponding type) in the program.

3-4
Navigate in the flow logic. Create a module for function code processing (using forward navigation) at PROCESS AFTER INPUT:

	Function code
	Action
	Next screen

	BACK
	None
	List

	SAVE
	First:
Information message no. 060(BC400)
	List

	All others
	None
	Screen 100


3-5
Ensure that pressing Enter always displays screen 100, regardless of the navigation history. Initialize the OK_CODE field in a PBO module.

Screens 

	Unit: Screens

Topic: Creating Screens


Model Solution: Program SAPBC400UDS_DYNPRO_A
*&--------------------------------------------------------------*

*& Report         SAPBC400UDS_DYNPRO_A                          *

*&                                                              *

*&                                                              *

*&--------------------------------------------------------------*

REPORT  sapbc400uds_dynpro_a.

CONSTANTS actvt_display TYPE activ_auth VALUE '03'.

* Definition of selection screen

PARAMETERS pa_anum TYPE sbook-agencynum.

* workarea for select

DATA: wa_booking TYPE sbc400_booking.

START-OF-SELECTION.

* selecting data using an ABAP dictionary view to get the data from
* sbook and the customer name from scustom
  SELECT carrid connid fldate bookid customid name

         FROM sbc400_booking

         INTO CORRESPONDING FIELDS OF wa_booking

           WHERE agencynum = pa_anum.

    AUTHORITY-CHECK OBJECT 'S_CARRID'

        ID 'CARRID' FIELD wa_booking-carrid

        ID 'ACTVT'  FIELD actvt_display.

    IF sy-subrc = 0.

* Output

      WRITE: / wa_booking-carrid COLOR col_key,

               wa_booking-connid COLOR col_key,

               wa_booking-fldate COLOR col_key,

               wa_booking-bookid COLOR col_key,

               wa_booking-name.

    ENDIF.

  ENDSELECT.

AT LINE-SELECTION.

  CALL SCREEN 100.
1-3
Create a screen using forward navigation

                  (double-click 100 in the CALL SCREEN 100 statement).

1-4
Maintain screen attributes

                  - Enter a descriptive short text

                  - Set the next screen to 0
1-5
Layout

                   - Navigate to the Graphical Layout Editor

                   - Choose the Dict/Program fields icon

                   - Enter SDYN_BOOK

                   - Choose the Get from Dictionary icon

                   - Select the fields you want

                     choose Enter to confirm, and drag these fields to the screen

                              - 1st Block:

           Copy key fields: CARRID, CONNID, FLDATE, and BOOKID
           with field names

                              - 2nd Block:

           Copy customer name NAME

           without a field name (choose the Without text radio button)

                              - 3rd Block:

           Copy the fields: CUSTOMID, CUSTTYPE, SMOKER, CLASS,       

           LOCCURAM, and LOCCURKEY
                                 with field names

1-6
Change the field attributes -

                   for example, by double-clicking the input field

                  - Change the attributes:

                  - The fields CARRID, CONNID, FLDATE, BOOKID, and CUSTOMID

                     should be displayed as output fields (Output field attribute).

           - The customer name NAME should be displayed next to the customer number


 without text         

                    (Output only attribute).

                  - The fields CUSTTYPE, SMOKER, CLASS, LOCCURAM, and 

                    LOCCURKEY should be both input-ready and output-ready (the Input/Output field 



attribute).

                        . 
Solutions

	[image: image46.png]



	Unit: Screens

Topic: Data Transport


Model Solution: Program SAPBC400UDS_DYNPRO_B
*&--------------------------------------------------------------*

*& Report         SAPBC400UDS_DYNPRO_B                          *
*&                                                              *

*&                                                              *

*&--------------------------------------------------------------*

REPORT  sapbc400uds_dynpro_b.
CONSTANTS: actvt_display TYPE activ_auth VALUE '03',
           actvt_change TYPE activ_auth VALUE '02'.
*  Definition of selection screen
PARAMETERS pa_anum TYPE sbook-agencynum.
* workarea for list
DATA wa_booking TYPE sbc400_booking.
* workarea for single booking to be changed
DATA wa_sbook TYPE sbook.
* workarea for dynpro
TABLES sdyn_book.
START-OF-SELECTION.
* selecting data using an Adictionary view to get the data from
* sbook and the customer name from scustom
  SELECT carrid connid fldate bookid customid name
         FROM sbc400_booking
         INTO CORRESPONDING FIELDS OF wa_booking
           WHERE agencynum = pa_anum.
    AUTHORITY-CHECK OBJECT 'S_CARRID'
        ID 'CARRID' FIELD wa_booking-carrid
        ID 'ACTVT'  FIELD actvt_display.
    IF sy-subrc = 0.
* Output
      WRITE: / wa_booking-carrid COLOR col_key,
               wa_booking-connid COLOR col_key,
               wa_booking-fldate COLOR col_key,
               wa_booking-bookid COLOR col_key,
               wa_booking-name.
      HIDE:    wa_booking-carrid,
               wa_booking-connid,
               wa_booking-fldate,
               wa_booking-bookid,
               wa_booking-name.
    ENDIF.
  ENDSELECT.
AT LINE-SELECTION.
  AUTHORITY-CHECK OBJECT 'S_CARRID'
          ID 'CARRID' FIELD wa_booking-carrid
          ID 'ACTVT'  FIELD actvt_change.
  IF sy-subrc = 0.
    SELECT SINGLE *
             FROM sbook
             INTO wa_sbook
             WHERE carrid = wa_booking-carrid
               AND connid = wa_booking-connid
               AND fldate = wa_booking-fldate
               AND bookid  = wa_booking-bookid.
    IF sy-subrc = 0.
      MOVE-CORRESPONDING wa_sbook TO sdyn_book.
      MOVE wa_booking-name TO sdyn_book-name.
      CALL SCREEN 100.
    ENDIF.
  ELSE .
    MESSAGE ID 'BC400' TYPE 'S' NUMBER '047'
    WITH wa_booking-carrid.
  ENDIF.

	Unit: Screens

Topic: Field Transports and Subsequent Screen Processing


Model Solution: Program SAPBC400UDS_DYNPRO_C
*&--------------------------------------------------------------*

*& Report         SAPBC400UDS_DYNPRO_C                          *
*&                                                              *

*&                                                              *

*&--------------------------------------------------------------*

REPORT  sapbc400uds_dynpro_c.
CONSTANTS: actvt_display TYPE activ_auth VALUE '03',
           actvt_change TYPE activ_auth VALUE '02'.
*  Definition of selection screen
PARAMETERS pa_anum TYPE sbook-agencynum.
* workarea for list
DATA wa_booking TYPE sbc400_booking.
* workarea for single booking to be changed
DATA wa_sbook TYPE sbook.
* workarea for dynpro
TABLES sdyn_book.
* variable for function code of user action
DATA ok_code LIKE sy-ucomm.
START-OF-SELECTION.
* selecting data using a dictionary view to get the data from
* sbook and the customer name from scustom
  SELECT carrid connid fldate bookid customid name
         FROM sbc400_booking
         INTO CORRESPONDING FIELDS OF wa_booking
           WHERE agencynum = pa_anum.
    AUTHORITY-CHECK OBJECT 'S_CARRID'
        ID 'CARRID' FIELD wa_booking-carrid
        ID 'ACTVT'  FIELD actvt_display.
    IF sy-subrc = 0.
* Output
      WRITE: / wa_booking-carrid COLOR col_key,
               wa_booking-connid COLOR col_key,
               wa_booking-fldate COLOR col_key,
               wa_booking-bookid COLOR col_key,
               wa_booking-name.
      HIDE:    wa_booking-carrid,
               wa_booking-connid,
               wa_booking-fldate,
               wa_booking-bookid,
               wa_booking-name.
    ENDIF.
  ENDSELECT.
  CLEAR wa_booking.
AT LINE-SELECTION.
  AUTHORITY-CHECK OBJECT 'S_CARRID'
         ID 'CARRID' FIELD wa_booking-carrid
         ID 'ACTVT'  FIELD actvt_change.
  IF sy-subrc = 0.
    SELECT SINGLE *
             FROM sbook
             INTO wa_sbook
             WHERE carrid = wa_booking-carrid
               AND connid = wa_booking-connid
               AND fldate = wa_booking-fldate
               AND bookid  = wa_booking-bookid.
    IF sy-subrc = 0.
      MOVE-CORRESPONDING wa_sbook TO sdyn_book.
      MOVE wa_booking-name TO sdyn_book-name.
      CALL SCREEN 100.
    ENDIF.
  ELSE .
    MESSAGE ID 'BC400' TYPE 'S' NUMBER '047'
    WITH wa_booking-carrid.
  ENDIF.
  CLEAR: wa_sbook, wa_booking.
*&--------------------------------------------------------------*

*&      Module  CLEAR_OK_CODE  OUTPUT                           *
*&--------------------------------------------------------------*

*&      text                                                    *
*&--------------------------------------------------------------*

MODULE clear_ok_code OUTPUT.
  CLEAR ok_code.
ENDMODULE.                 " CLEAR_OK_CODE  OUTPUT
*&--------------------------------------------------------------*

*&      Module  USER_COMMAND_0100  INPUT                        *
*&--------------------------------------------------------------*

*&      text                                                    *
*&--------------------------------------------------------------*

MODULE user_command_0100 INPUT.
  CASE ok_code.
    WHEN 'BACK'.
      LEAVE TO SCREEN 0.
    WHEN 'SAVE'.
      MOVE-CORRESPONDING sdyn_book TO wa_sbook.
      MESSAGE ID 'BC400' TYPE 'I' NUMBER '060'.
      LEAVE TO SCREEN 0.
  ENDCASE.
ENDMODULE.                             " USER_COMMAND_0100  INPUT

